

FUNCTIONAL PEARL

Formatting: a class act

RALF HINZE

*Institute of Information and Computing Sciences, Utrecht University
 P.O.Box 80.089, 3508 TB Utrecht, The Netherlands
 (e-mail: ralf@cs.uu.nl)*

1 Introduction

When I was a student, Simula was one of the languages taught in introductory programming language courses and I vividly remember a sticker one of our instructors had attached to the door of his office, saying “Simula does it with class”. I guess the same holds for Haskell except that Haskell replaces classes by type classes.

Armed with singleton types, multiple-parameter type classes, and functional dependencies we reconsider a problem raised and solved by Danvy in a previous pearl (1998). The challenge is to implement a variant of C’s *printf* function, called *format* below, in a statically typed language. Here is an interactive session that illustrates the problem:

```
Main> :type format (lit "hello world")
String
Main> format (lit "hello world")
"hello world"
Main> :type format int
Int → String
Main> format int 5
"5"
Main> :type format (int ^ lit " is " ^ str)
Int → String → String
Main> format (int ^ lit " is " ^ str) 5 "five"
"5 is five".
```

The *format* directive *lit s* means emit *s* literally. The directives *int* and *str* instruct *format* to take an additional argument of the types *Int* and *String* respectively, which is then shown. The circumflex ‘ \wedge ’ is used to concatenate two directives.

The type of *format* depends on its first argument, the *format* directive. In a language with dependent types, such as Cayenne (Augustsson, 1999), *format* is straightforward to implement. This pearl shows that *format* is equally straightforward to realize in a language like Haskell that allows the definition of values that depend on types. Our solution enjoys nice algebraic properties and is more direct than Danvy’s one (the relation between the two approaches is detailed in Sec. 5).

2 Preliminaries: functors

This section briefly reviews the categorical concept of a functor, which is at the heart of the Haskell implementation of *format*. For our purposes it is sufficient to think of a *functor* as a combination of a type constructor F of kind $\star \rightarrow \star$ and a so-called *mapping function* that lifts a given function of type $A \rightarrow B$ to a function of type $F A \rightarrow F B$. In Haskell, the concept of a functor is captured by the following class definition:¹

```
class Functor F where
  map :: (A → B) → (F A → F B).
```

Instances of this class are supposed to satisfy the two *functor laws*:

$$\begin{aligned} \text{map } id &= id \\ \text{map } (\phi \cdot \psi) &= \text{map } \phi \cdot \text{map } \psi. \end{aligned}$$

Typical examples of functors are container types such as lists or trees. In these cases, the mapping function simply applies its first argument to each element of a given container, leaving its structure intact. However, the notion of functor is by no means restricted to container types. For instance, the functional type $(A \rightarrow)$ for fixed A is a functor with the mapping function given by *post-composition*.²

```
instance Functor (A →) where
  map φ x = φ · x
```

For this instance, the functor laws reduce to $id \cdot x = x$ and $(\phi \cdot \psi) \cdot x = \phi \cdot (\psi \cdot x)$. The functor $(A \rightarrow)$ will play a prominent rôle in the following sections. In addition, we require the *identity functor* and *functor composition*.

```
type Id A = A
instance Functor Id where
  map = id
type (F · G) A = F (G A)
instance (Functor F, Functor G) ⇒ Functor (F · G) where
  map = map · map
```

Again, it is easy to see that the functor laws are satisfied. Furthermore, functor composition is associative and has the identity functor as a unit. As an aside, note that these instance declarations are not legal Haskell since *Id* and ‘·’ are not data types defined by **data** or by **newtype**. A data type, however, introduces an additional data constructor which affects the readability of the code. Instead we employ **type** declarations as if they worked as **newtype** declarations. Sec. 6 describes the necessary amendments to make the code run under GHC or Hugs.

¹ We slightly deviate from Haskell’s lexical syntax: both type constructors and type variables are written with an initial upper-case letter (a type variable typically consists of a single upper-case letter) and both value constructors and value variables are written with an initial lower-case letter. This convention helps us to keep values and types apart.

² The so-called *operator section* $(A \rightarrow)$ denotes the partial application of the infix operator ‘ \rightarrow ’ to A .

3 Functional unparsing

The Haskell solution is developed in two steps. In this section we show how to define *format* as a *type-indexed value*. The following section then explains how to implement the type-indexed value using multiple-parameter type classes with functional dependencies.

Recall that the type of *format* depends on its first argument, the format directive. Clearly, we cannot define such a dependently typed function in Haskell if we represent directives by elements of a single data type, say,

```
data Dir = lit String | int | str | Dir ^ Dir.
```

However, using Haskell's type classes we can define values that depend on types. In order to utilize this feature we must arrange that each directive possesses a distinct type. To this end we introduce the following *singleton types*:

```
data LIT    = lit String
data INT   = int
data STR   = str
data D1 ^ D2 = D1 ^ D2.
```

Strictly speaking, *LIT* is not a singleton type since it accommodates more than one element. This is unproblematic, however, since the type of *format* does not depend on the argument of *lit*. Given these declarations, the directive *int ^ lit " is " ^ str*, for instance, has type *INT ^ LIT ^ STR*: the structure of the directive is mirrored at the type level. As an aside, note that the type constructor ‘ \wedge ’, which takes singleton types to singleton types, is isomorphic to the type of pairs. We could have used pairs in the first place but the right-associative infix data constructor ‘ \wedge ’ saves some parentheses.

We can now define *format* as a type-indexed value of type

```
format_D :: D → Format_D String.
```

That is, *format_D* takes a directive of type *D* and returns ‘something’ of *String* where ‘something’ is determined by *D* in the following way:

```
Format_{D::*}    :: * → *
Format_{LIT} S   = S
Format_{INT} S   = Int → S
Format_{STR} S   = String → S
Format_{D_1 ^ D_2} S = Format_{D_1} (Format_{D_2} S).
```

The type *Format_D* is a so-called *type-indexed type*, a type that depends on a type. It specifies for each of the directives the additional argument(s) *format* has to take. The most interesting clause is probably the last one: the arguments to be added for *D₁ ^ D₂* are the arguments to be added for *D₁* followed by the arguments to be added for *D₂*. The crucial property of *Format_D* is that it constitutes a functor.

This can be seen more clearly if we rewrite Format_D in a point-free style.

$$\begin{aligned}\text{Format}_{\text{LIT}} &= \text{Id} \\ \text{Format}_{\text{INT}} &= (\text{Int} \rightarrow) \\ \text{Format}_{\text{STR}} &= (\text{String} \rightarrow) \\ \text{Format}_{D_1 \cdot D_2} &= \text{Format}_{D_1} \cdot \text{Format}_{D_2}\end{aligned}$$

The implementation of format is straightforward except perhaps for the last case.

$$\begin{aligned}\text{format}_D &\quad :: D \rightarrow \text{Format}_D \text{ String} \\ \text{format}_{\text{LIT}} (\text{lit } s) &= s \\ \text{format}_{\text{INT}} \text{ int} &= \lambda i \rightarrow \text{show } i \\ \text{format}_{\text{STR}} \text{ str} &= \lambda s \rightarrow s \\ \text{format}_{D_1 \cdot D_2} (d_1 \cdot d_2) &= \text{format}_{D_1} d_1 \diamond \text{format}_{D_2} d_2\end{aligned}$$

So $\text{format}_{\text{INT}} \text{ int}$ is just the show function and $\text{format}_{\text{STR}} \text{ str}$ is just the identity on String . It remains to define the operator ‘ \diamond ’, which takes an $F \text{ String}$ and a $G \text{ String}$ to a $(F \cdot G) \text{ String}$. We know that $F = \text{Format}_{D_1}$ and $G = \text{Format}_{D_2}$ but this does not get us any further. The only assumption we may safely take is that F and G are functorial. Fortunately, using the mapping function on F we can turn a value of type $F \text{ String}$ into a value of type $F(G \text{ String})$ provided we supply a function that takes a String , say, s to a value of type $G \text{ String}$. We can define a function of the desired type using the mapping function on G provided we supply a function that takes a string, say, t to some resulting string. Now, since we have to concatenate the ‘output’ produced by the two arguments of ‘ \diamond ’, the resulting string must be $s + t$.

$$\begin{aligned}\diamond &\quad :: (\text{Functor } F, \text{Functor } G) \Rightarrow F \text{ String} \rightarrow G \text{ String} \rightarrow (F \cdot G) \text{ String} \\ f \diamond g &= \text{map}_F (\lambda s \rightarrow \text{map}_G (\lambda t \rightarrow s + t) g) f\end{aligned}$$

The operator ‘ \diamond ’ enjoys nice algebraic properties: it is associative and has the empty string, “” :: Id String , as a unit. The proof of these properties makes use of the functor laws and the fact that $(\text{String}, +, "")$ forms a monoid. That said it becomes clear that the construction can be readily generalized to arbitrary monoids. As an example, for reasons of efficiency one might want to replace $(\text{String}, +, "")$ by $(\text{ShowS}, \cdot, \text{id})$, which features constant-time concatenation.

4 Functional unparsing in Haskell

How can we implement the type-indexed value $\text{format}_D :: D \rightarrow \text{Format}_D \text{ String}$ using Haskell’s type classes? Clearly, a singleton parameter class won’t do since both D and Format_D vary. We are forced to introduce a two argument class that additionally abstracts away from Format_D assigning format the general type $D \rightarrow F \text{ String}$. This type is, however, too general since now D and F may vary independently of each other. This additional ‘flexibility’ is, in fact, not very welcome since it gives rise to severe problems of ambiguity. Fortunately, *functional dependencies* (Jones,

2000) save the day as they allow us to capture the fact that F is determined by D .

```
class (Functor F)  $\Rightarrow$  Format D F | D  $\rightarrow$  F where
  format :: D  $\rightarrow$  F String
```

The functional dependency $D \rightarrow F$ (beware, this is not the function space arrow) constrains the relation to be functional: if there are instances $\text{Format } D_1 F_1$ and $\text{Format } D_2 F_2$, then $D_1 = D_2$ implies $F_1 = F_2$. Note that F is additionally restricted to be an instance of *Functor*. It remains to supply for each directive D an instance declaration of the form **instance** $\text{Format } D (\text{Format}_D)$ **where** $\text{format} = \text{format}_D$.

```
instance Format LIT Id where
  format (lit s) = s
instance Format INT (Int →) where
  format int =  $\lambda i \rightarrow \text{show } i$ 
instance Format STR (String →) where
  format str =  $\lambda s \rightarrow s$ 
instance (Format D1 F1, Format D2 F2)  $\Rightarrow$  Format (D1 ^ D2) (F1 · F2) where
  format (d1 ^ d2) = format d1 ◊ format d2
```

In implementing the specification of Sec. 3 we have simply replaced a type function by a functional type relation. Before we proceed let us take a look at an example translation.

```
format (int ^ lit " is " ^ str)
= { definition of format }
  show ◊ " is " ◊ id
= { definition of '◊' }
   $\text{map}_{\text{Int} \rightarrow} (\lambda s \rightarrow \text{map}_{\text{Id}} (\lambda t \rightarrow \text{map}_{\text{String} \rightarrow} (\lambda u \rightarrow s ++ t ++ u) \text{id}) " is ") \text{ show}$ 
= { definition of  $\text{map}_{A \rightarrow}$  and  $\text{map}_{\text{Id}}$  }
   $(\lambda s \rightarrow (\lambda t \rightarrow (\lambda u \rightarrow s ++ t ++ u) \cdot \text{id}) " is ") \cdot \text{show}$ 
= { algebraic simplifications and  $\beta$ -conversion }
   $\lambda i \rightarrow \lambda u \rightarrow \text{show } i ++ " is " ++ u$ 
```

We obtain exactly the function one would have written by hand. Note that simplifications along these lines can always be performed at compile time since the first argument of *format* is essentially static (apart from *lit*'s string argument).

5 Back to continuation-passing style

It is instructive to compare our solution to the original one by Danvy (1998), which makes use of a *continuation* and an *accumulating argument*. Phrased as a Haskell type class, Danvy's solution reads:

```
class Format' D F | D  $\rightarrow$  F where
  format' ::  $\forall A. D \rightarrow (\text{String} \rightarrow A) \rightarrow (\text{String} \rightarrow F A)$ 
```

```

instance Format' LIT Id where
  format' (lit s) =  $\lambda\kappa\ out \rightarrow \kappa\ (out + s)$ 
instance Format' INT (Int  $\rightarrow$ ) where
  format' int =  $\lambda\kappa\ out \rightarrow \lambda i \rightarrow \kappa\ (out + show\ i)$ 
instance Format' STR (String  $\rightarrow$ ) where
  format' str =  $\lambda\kappa\ out \rightarrow \lambda s \rightarrow \kappa\ (out + s)$ 
instance (Format' D1 F1, Format' D2 F2)  $\Rightarrow$  Format' (D1  $\wedge$  D2) (F1  $\cdot$  F2) where
  format' (d1  $\wedge$  d2) =  $\lambda\kappa\ out \rightarrow format'\ d1\ (format'\ d2\ \kappa)\ out$ 
  format :: (Format' D F)  $\Rightarrow$  D  $\rightarrow$  F String
  format d = format' d id "".

```

Two remarks are in order. First, the instances do not require mapping functions, which explains why *Format'* is not declared a subclass of *Functor*, though morally the second argument of *Format'* is a functor. Second, *format'* ($d_1 \wedge d_2$) can be simplified to *format'* $d_1 \cdot format'$ d_2 , where ‘ \cdot ’ is ordinary function composition. We will take these points up again in the following section.

So we interpret *format d* by values of type *F String* whereas Danvy employs values of type $\forall A.(A \rightarrow String) \rightarrow (A \rightarrow F String)$. An obvious question is, of course, whether the two approaches are equivalent. Here are functions that convert to and fro:

$$\begin{aligned}\alpha\ d &= \lambda\kappa\ out \rightarrow map\ (\lambda s \rightarrow \kappa\ (out + s))\ d \\ \gamma\ d' &= d' id\ "".\end{aligned}$$

The coercion function α introduces a continuation and an accumulating string, while γ supplies an initial continuation and an empty accumulating string.

It is easy to see that $\gamma \cdot \alpha = id$:

$$\begin{aligned}\gamma(\alpha d) &= \{ \text{definition of } \gamma \text{ and } \alpha \} \\ &\quad (\lambda\kappa\ out \rightarrow map\ (\lambda s \rightarrow \kappa\ (out + s))\ d) id\ "" \\ &= \{ \beta\text{-conversion} \} \\ &\quad map\ (\lambda s \rightarrow id\ (""\ + s))\ d \\ &= \{ \text{algebraic simplifications} \} \\ &\quad map\ id\ d \\ &= \{ \text{functor laws} \} \\ &\quad d.\end{aligned}$$

When we try to prove the converse, $\alpha \cdot \gamma = id$,

$$\begin{aligned}\alpha(\gamma d') &= \{ \text{definition of } \alpha \text{ and } \gamma \} \\ &\quad \lambda\kappa\ out \rightarrow map\ (\lambda s \rightarrow \kappa\ (out + s))\ (d' id\ ""),\end{aligned}$$

we are immediately stuck. There is no obvious way to simplify the final expression. Note, however, that d' has a polymorphic type, so we can appeal to the *parametricity theorem* (Wadler, 1989). The ‘free theorem’ for $d' :: \forall A.(String \rightarrow A) \rightarrow (String \rightarrow F A)$ is that for all $\phi :: A_1 \rightarrow A_2$ and for all $\epsilon :: String \rightarrow A_1$,

$$map\ \phi \cdot d' \epsilon = d'(\phi \cdot \epsilon). \tag{1}$$

Loosely speaking, this rule allows us to shift a part of the continuation to the left. Continuing the proof we obtain:

$$= \{ \text{parametricity (1): } \phi = \lambda s \rightarrow \kappa (\text{out} ++ s) \text{ and } \epsilon = \text{id} \}$$

$$\lambda \kappa \text{ out} \rightarrow d' (\lambda s \rightarrow \kappa (\text{out} ++ s)) "".$$

We are stuck again. This time we require a rule that allows us to shift a part of the continuation to the right. Let us assume for the moment that for all $\epsilon :: \text{String} \rightarrow A$ and for all $\sigma :: \text{String} \rightarrow \text{String}$,

$$d' (\epsilon \cdot \sigma) = d' \epsilon \cdot \sigma. \quad (2)$$

Given this property, we can finish the proof:

$$= \{ \text{proof obligation (2): } \epsilon = \kappa \text{ and } \sigma = \lambda s \rightarrow \text{out} ++ s \}$$

$$\lambda \kappa \text{ out} \rightarrow d' (\lambda s \rightarrow \kappa s) (\text{out} ++ "")$$

$$= \{ \text{algebraic simplifications and } \eta\text{-conversion} \}$$

$$d'.$$

It remains to establish the proof obligation. Perhaps unsurprisingly, it turns out that the rule does not hold in general. The problem is that the accumulating argument has a too concrete type: it is a string, which we can manipulate at will. In the following instance, for example, the accumulator is replaced by an empty string.

```
data CANCEL = cancel
instance Format' CANCEL Id where
    format' cancel = λκ out → κ ""
```

The effect of *cancel* is to discard the string produced by the directives to its left.

```
Main> format' (int ^ lit " is " ^ cancel ^ str) 5 "five"
"five"
```

One might argue that the ability to define such a directive is an unwanted consequence of switching to continuation passing style. In that sense, rule (2) is really a proof obligation for the programmer. As a closing remark, note that we can achieve a similar effect in our setting using a ‘forgetful’ variant of ‘ \diamond ’:

$$f \triangleright g = \text{map} (\lambda s \rightarrow \text{map} (\lambda t \rightarrow t) g) f$$

$$f \triangleleft g = \text{map} (\lambda s \rightarrow \text{map} (\lambda t \rightarrow s) g) f.$$

6 Applying a functor

Let us finally turn the code of Sec. 4 into an executable Haskell program (for consistency, we stick to the lexical conventions of Sec. 2). Recall that the instance declarations involving the type synonyms *Id* and ‘ \cdot ’ are not legal since type synonyms must not be partially applied. Therefore, we are forced to introduce the two types via **newtype** declarations:

```
newtype Id A = ide A
newtype (F · G) A = com (F (G A)).
```

Alas, now Id and ‘ \cdot ’ are new distinct types. In particular, the identities $Id\ A = A$ and $(F \cdot G)\ A = F\ (G\ A)$ do not hold any more: the type of $format$ ($int \diamond lit$ “is” $^{\wedge} str$) is $((Int \rightarrow) \cdot Id \cdot (String \rightarrow))\ String$ rather than $Int \rightarrow String \rightarrow String$. In order to obtain the desired type we have to apply the functor $(Int \rightarrow) \cdot Id \cdot (String \rightarrow)$ to the type $String$. This type transformation is implemented by the following three parameter type class:

```
class (Functor F)  $\Rightarrow$  Apply F A B | F A  $\rightarrow$  B where
  apply  $\quad :: F\ A \rightarrow B$ 
instance Apply (A  $\rightarrow$  B) (A  $\rightarrow$  B) where
  apply  $\quad = id$ 
instance Apply Id A A where
  apply (ide a)  $= a$ 
instance (Apply G A B, Apply F B C)  $\Rightarrow$  Apply (F  $\cdot$  G) A C where
  apply (com x)  $= apply (map\ apply\ x).$ 
```

The intention is that the type relation $Apply\ F\ A\ B$ holds iff $F\ A = B$. Consequently, B is uniquely determined by F and A , which is expressed by the functional dependency $F\ A \rightarrow B$ (again, do not confuse the dependency with a functional type). The class method $apply$ always equals the identity function since a **newtype** has the same representation as the underlying type. Now, renaming the class method of $Format$ to $formatx$ we arrive at the true definition of $format$:

```
format  $\quad :: (Format\ D\ F, Apply\ F\ String\ A) \Rightarrow D \rightarrow A$ 
format d  $= apply (formatx\ d).$ 
```

7 Haskell can do it (almost) without type classes

Given the title of the pearl this final twist is perhaps unexpected. We can quite easily eliminate the $Format$ class by specializing $format$ to the various types of directives: for each $d :: D$ we introduce a new directive $\underline{d} :: Format_D\ String$ given by $\underline{d} = formatx\ d$ —we omit the underlining in the sequel and just reuse the original names.

<i>lit</i>	$:: String \rightarrow Id\ String$
<i>lit s</i>	$= ide\ s$
<i>int</i>	$:: (Int \rightarrow) String$
<i>int</i>	$= \lambda i \rightarrow show\ i$
<i>str</i>	$:: (String \rightarrow) String$
<i>str</i>	$= \lambda s \rightarrow s$
<i>format</i>	$:: (Apply\ F\ String\ A) \Rightarrow F\ String \rightarrow A$
<i>format d</i>	$= apply\ d$
<i>formatIO</i>	$:: (Apply\ F\ (IO\ ())\ A) \Rightarrow F\ String \rightarrow A$
<i>formatIO d</i>	$= apply\ (map\ putStrLn\ d)$

So int is just $show$ (albeit with a less general type), str is just id , and $format$ is just $apply$ (again with a less general type). Furthermore, instead of ‘ \diamond ’ we use ‘ \diamond ’.

We have also defined a variant of *format* that outputs the string to the standard output device. This function nicely demonstrates how to define one's own variable-argument functions on top of *format*. Here is an example session that illustrates the use of the new unparsing combinators:

```
Main> :type (int `diamond` lit " is " `diamond` str)
((Int ->) · Id · (String ->)) String
Main> :type format (int `diamond` lit " is " `diamond` str)
Int -> String -> String
Main> format (int `diamond` lit " is " `diamond` str) 5 "five"
"5 is five"
Main> format (show `diamond` lit " is " `diamond` show) 5 "five"
"5 is \"five\""
Main> format (lit "sum " `diamond` show `diamond` lit " = " `diamond` show) [1..10] (sum [1..10])
"sum [1,2,3,4,5,6,7,8,9,10] = 55".
```

Note the use of *show* in the last two examples. In fact, we can now seamlessly integrate Haskell's predefined unparsing function with our own routines. As an illustration, consider the following directive for unparsing a list of values:

<i>list</i>	$\text{:: } (A \rightarrow) \text{String} \rightarrow ([A] \rightarrow) \text{String}$
<i>list d []</i>	$= "[]"$
<i>list d (a : as)</i>	$= "[" ++ d a ++ rest as$
where <i>rest []</i>	$= "]"$
<i>rest (a : as)</i>	$= ", " ++ d a ++ rest as.$

To format a string, for instance, we can now either use the directive *str* (emit the string literally), *show* (put the string in quotes), or *list show* (show the string as a list of characters). Likewise, for formatting a list of strings we can choose between *show*, *list str*, *list show*, or *list (list show)*.

Can we also get rid of *Id*, ‘·’ and consequently of the class *Apply*? Unfortunately, the answer is in the negative. Though all directives possess legal Haskell 98 types, Haskell's kinded first-order unification gets in the way when we combine the directives. Loosely speaking, the **newtype** constructors are required to direct the type checker. Interestingly, Danvy's solution seems to require a less sophisticated type system: the combinators possess ordinary Hindley-Milner types. However, this comes at the expense of type safety as a closer inspection reveals. The critical combinator is the one for concatenating directives, which possesses the following rank-2 type (consider the instance declaration for ‘^’ in Sec. 5):

$$\begin{aligned} (\cdot) &\text{ :: } \forall F \ G . (\forall X . (\text{String} \rightarrow X) \rightarrow (\text{String} \rightarrow F \ X)) \\ &\quad \rightarrow (\forall Y . (\text{String} \rightarrow Y) \rightarrow (\text{String} \rightarrow G \ Y)) \\ &\quad \rightarrow (\forall Z . (\text{String} \rightarrow Z) \rightarrow (\text{String} \rightarrow F \ (G \ Z))). \end{aligned}$$

Since ‘·’ amounts to function composition we can generalize (or rather, weaken) the

type to

$$\begin{aligned} (\cdot) :: \forall A B C . & (B \rightarrow C) \\ & \rightarrow (A \rightarrow B) \\ & \rightarrow (A \rightarrow C). \end{aligned}$$

Since Danvy's combinators furthermore do not employ mapping functions, they can be made to run in a language with a Hindley-Milner type system. Of course, weakening the types has the immediate drawback that, for instance, the non-sensible call *format* (*const* · *length* · *run*) where *run k = k ""* is well-typed.

8 Summary and further reading

Do you have a similar problem that involves capturing dependent types in Haskell? Here is a brief summary of the overall construction. Let us assume that we are given a dependently typed function $f :: (t :: T) \rightarrow F t$ where $F :: T \rightarrow \star$ is the dependent type. The central step is to lift the elements of T to the type level such that for every element t of T there is a corresponding type \bar{t} . Through the lifting we obtain a family of functions $f_{\bar{t}} :: \bar{t} \rightarrow F_{\bar{t}}$ and a family of types $F_{\bar{t}} :: \star$ both indexed by type. These families can be immediately represented in Haskell using a two argument type class with a functional dependency. If the function f is defined by structural recursion, then each equation gives rise to an instance declaration (the corresponding equation of the dependent type F goes into the instance head albeit in a relational form).

Do you want to delve deeper into the world of singleton types, multiple-parameter type classes, and functional dependencies? In this case, I recommend reading (Hallgren, 2001; Neubauer *et al.*, 2001; Gasbichler *et al.*, 2002; McBride, 2002).

9 Acknowledgements

Thanks are due to Dave Clarke, Johan Jeuring, Andres Löh, Doaitse Swierstra, Peter Thiemann, and Stephanie Weirich for their comments on a previous draft of this paper.

References

- Augustsson, L. (1999) Cayenne – a language with dependent types. *SIGPLAN Notices* **34**(1):239–250.
- Danvy, O. (1998) Functional unparsing. *J. Functional Programming* **8**(6):621–625.
- Gasbichler, M., Neubauer, M., Sperber, M. and Thiemann, P. (2002) Functional logic overloading. *Proceedings of the 29th Annual ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages (POPL' 02), Portland, Oregon, January 16–18.*
- Hallgren, T. (2001) Fun with functional dependencies. *Proceedings of the Joint CS/CE Winter Meeting*, pp. 135–145. Department of Computing Science, Chalmers, Göteborg, Sweden.
- Jones, M. P. (2000) Type classes with functional dependencies. Smolka, G. (ed), *Proceedings of the 9th European Symposium on Programming, ESOP 2000, Berlin, Germany*. Lecture Notes in Computer Science 1782, pp. 230–244. Springer-Verlag.

- McBride, C. (2002) Faking it (simulating dependent types in Haskell). *J. Functional Programming*. to appear.
- Neubauer, M., Thiemann, P., Gasbichler, M. and Sperber, M. (2001) A functional notation for functional dependencies. Hinze, R. (ed), *Proceedings of the 2001 ACM SIGPLAN Haskell Workshop, Firenze, Italy*, pp. 101–120. The preliminary proceedings appeared as a University of Utrecht technical report, UU-CS-2001-23.
- Wadler, P. (1989) Theorems for free! *The Fourth International Conference on Functional Programming Languages and Computer Architecture (FPCA'89), London, UK*, pp. 347–359. Addison-Wesley Publishing Company.